


# Onion Domes: Some Amendments

H Nooshin and O A Samavati, 5<sup>th</sup> February 2015

Section 3.4.2 of FCP (Formex Configuration Processing) is concerned with the formex formulation of ‘onion domes’ using a ‘toroidal normat (toroidal coordinate system)’. A cross-section of a toroidal normat is shown in FCP, Fig 3.4.22. This figure is reproduced here:




The part of the normat shown in thick line is the region of the normat for creation of onion domes. The symbols used in the figure are as follows:

- U1, U2, U3 and U4 represent the toroidal directions (coordinates).
- R is the ‘first radius’ of the toroidal normat (in the U1 direction).
- Q is the ‘second radius’ of the toroidal normat (in the U3 direction).
- A is the ‘sweep angle’ of the onion dome.
- B is the ‘base angle’ of the onion dome.
- P is the ‘position angle’ of the onion dome, indicating the position of the ‘crown’ of the dome with respect to the U3 direction.

**Note:** It is suggested that the term ‘position angle’ is changed to ‘crown angle’ and the letter C rather than P is used to represent it.

The ‘curviance’ of an onion dome can be specified in terms of the parameters R, Q and B. Also, it is convenient to include a ‘vertical scaling factor’ (represented by V) for the adjustment of the proportions of the dome.



**Amendments:** The amendments contained in the present document are concerned with the values of Span (S), Diameter (D) and Rise (H) of the dome. These are shown in the sketch above:

Note that S represents the diameter of the bottom opening of the dome, whereas D represents the diameter of the largest girth of the dome.

The most straightforward strategy for formulation of the geometry of an onion dome is in terms of the variables R, Q, B and V. On the other hand, values of span and rise are important design parameters for any dome. These values together with the diameter of an onion dome are formulated as follows:

The diameter D of the dome is given by:

$$D = 2 (Q - R)$$

Note that in a toroidal normal suitable for creation of onion domes, the second radius Q is always larger than the first radius R.

The span S of the dome may be formulated as follows:

$$S = D - 2 (Q - Q \cos B)$$

or

$$S = 2 (Q - R) - 2 (Q - Q \cos B)$$

or

$$S = 2 (Q \cos B - R)$$

Thus, if the base angle  $B = 0$ , then  $S = D$ . Note that if  $B < 0$  then the dome has no 'diameter', as such.

Finally, the rise H of the dome is given by:

$$H = V (Q \cos P + Q \sin B)$$

or

$$H = V Q (\cos P + \sin B)$$

A modified version of the Formian scheme (program) in Fig 3.4.23 of the FCP, including the formulations of the Span, Diameter and Rise is shown below:

(\*) Onion dome of Fig 3.4.19 (A Formian-K scheme) (\*)

(\*) Setting parameters (\*)

R=10; (\*) First radius (\*)

Q=50; (\*) Second radius (\*)

B=40; (\*) Base angle (\*)

m=24; (\*) Number of rhombi along U2 (\*)

n=6; (\*) Number of rhombi along U4 (\*)

V=1; (\*) Vertical scale factor (\*)

A=acos|(R/Q)+B; (\*) Sweep angle (\*)

P=asin|(R/Q); (\*) Position angle (\*)

(\*)Formulating the configuration in toroidal coordinates (\*)

E=rinic(2,4,m,n,2,-2)|lamic(2,4,1,-1)|[1,1,1,0; 1,0,1,-1]#

rin(2,m,2)|[1,1,1,-2\*n; 1,3,1,-2\*n];

```
F=tran(4,-2*n*P/A)|E;
```

(\*) Transferring to the global Cartesian coordinates (\*)

```
G1=bto(R,360/(2*n),Q,A/(2*n))|F;
```

(\*) Note that here the name 'basitoroidal' (bto) rather than 'basiannular' (ba) is used (\*)

(\*) Removing the unwanted fourth direction (\*)

```
G2=dep(4)|G1;
```

(\*) Applying scale and removing the repeated elements (\*)

```
Dome=pex|dil(3,V)|G2;
```

```
Span=2*(Q*cos|B - R); Diameter=2*(Q - R);
```

```
Rise=V*Q*(cos|P+sin|B);
```

(\*) Setting view particulars (\*)

```
use &,vm(2),vt(2),vh(0,10000*Q,0, 0,0,0, 0,0,1);
```

```
clear; draw Dome;
```

```
erase; give Span, Diameter, Rise;
```

◇